
Περιεχόµενα

• Όροι (terms)

• Χρήση ενός συστήµατος Prolog

• Κανόνες (rules)

• Οι δύο όψεις της Prolog

• Τελεστές (operators)

• Λίστες (lists)

• Άρνηση ως αποτυχία (negation as failure)

• Σε τί είναι καλή η Prolog;

Όροι (Terms)

• Κάθε τι στην Prolog κατασκευάζεται από όρους:

– Τα προγράµµατα σε Prolog

– Τα δεδοµένα που χειρίζονται τα προγράµµατα σε Prolog

• Τρεις κατηγορίες όρων:

– ΜεταβλητέςΜεταβλητέςΜεταβλητέςΜεταβλητές

– ΣταθερέςΣταθερέςΣταθερέςΣταθερές: ακέραιοι (integers), πραγµατικοί αριθµοί (reals), και – ΣταθερέςΣταθερέςΣταθερέςΣταθερές: ακέραιοι (integers), πραγµατικοί αριθµοί (reals), και 
άτοµα (atoms)

– Σύνθετοι όροιΣύνθετοι όροιΣύνθετοι όροιΣύνθετοι όροι: λίστες (lists) και δοµηµένα δεδοµένα (structures)

Εισαγωγή στη γλώσσα Prolog 3

Άτοµα (Atoms)

• Ένα άτοµο είναι µια σταθερά που αποτελείται από 

αλφαριθµητικούς χαρακτήρες:

– an , atom , has , many_characters

• Εάν ο αρχικός χαρακτήρας δεν είναι ένα µικρό γράµµα ή 

υπάρχουν «περίεργοι» χαρακτήρες, το άτοµο πρέπει να υπάρχουν «περίεργοι» χαρακτήρες, το άτοµο πρέπει να 

βρίσκεται µέσα σε αποστρόφους

– ’Kostis’ , ’40 παλικάρια’ , ’$^%(@(&$#*&$’

• Κάθε άτοµο έχει µια τιµή (τους χαρακτήρες του) και 

είναι ίσο µόνο µε άλλα άτοµα που έχουν ακριβώς τους 

ίδιους χαρακτήρες ως τιµή

• Σκεφτείτε τα άτοµα κάτι σαν strings σε άλλες γλώσσες

Μεταβλητές

• Μεταβλητή είναι κάθε τι που αρχίζει από ένα κεφαλαίο 

γράµµα ή underscore, και ακολουθείται από γράµµατα, 

αριθµούς ή underscores:

_ , X, Variable , Kostis , _123

• Οι µεταβλητές έχουν δύο καταστάσεις:

– Μπορεί να µην έχουν ακόµα δεθεί µε κάποια τιµή (unbound)

– Μπορεί να έχουν κάποια τιµή, δηλαδή να είναι δεµένες (bound)

• Όταν όµως µια µεταβλητή είναι σε κατάσταση «µε τιµή», 

η τιµή αυτή δεν αλλάζει

Σύνθετοι όροι (Compound terms)

• Αρχίζουν µε ένα άτοµο και ακολουθούνται από µια 

ακολουθία όρων που διαχωρίζονται µε κόµµατα και 

περικλείονται από παρενθέσεις:

– university(' Ε. Μ. Π.')

– ' +' ( 2, 3.14 )– ' +' ( 2, 3.14 )

– couple(adam,eve)

– tree(V,tree(42,Y,Z),R)

• Οι λίστες είναι και αυτές σύνθετοι όροι αλλά όπως θα 

δούµε έχουν ειδική σύνταξη

• Σκεφτείτε τους σύνθετους όρους σαν δοµές δεδοµένων

Σύνταξη των όρων

<term> ::= <constant> | <variable> | <compound-term>
<constant> ::= <integer> | <real number> | <atom>
<compound-term> ::= <atom> ( <termlist> )
<termlist> ::= <term> | <term> , <termlist>

• Όλα τα προγράµµατα και τα δεδοµένα της Prolog 

κατασκευάζονται από τέτοιους όρους

• Σε λίγο θα δούµε ότι, για παράδειγµα, ο σύνθετος όρος

'+'(1,2) συνήθως γράφεται ως 1+2

• Αλλά αυτή είναι απλά µια συντοµογραφία του ίδιου όρου

Ενοποίηση (Unification)

• Ταίριασµα προτύπων σε όρους της Prolog

• ∆ύο όροι ενοποιούνται εάν υπάρχει κάποιος τρόπος να 

βρεθούν τιµές για τις µεταβλητές τους που να τους 

κάνουν ακριβώς ίδιους

• Για παράδειγµα, οι όροι couple( adam,eve ) και• Για παράδειγµα, οι όροι couple( adam,eve ) και

couple(Who,WithWhom) ενοποιούνται αν η µεταβλητή 

Whoπάρει την τιµή adam και η µεταβλητή WithWhom

πάρει την τιµή eve

• (Περισσότερες λεπτοµέρειες στο επόµενο µάθηµα…)

Η βάση δεδοµένων της Prolog

• Η υλοποίηση της Prolog διατηρεί µια αναπαράσταση των 

προγραµµάτων που έχουν φορτωθεί στο σύστηµα

• Ο κώδικας συµπεριφέρεται σαν µια βάση δεδοµένων

• Ένα πρόγραµµα Prolog είναι απλώς ένα σύνολο από 

δεδοµένα για αυτή τη βάσηδεδοµένα για αυτή τη βάση

• Το απλούστερο από τα δεδοµένα αυτής της βάσης είναι 

ένα γεγονός (fact): συντακτικά, ένα γεγονός είναι ένας 

όρος που ακολουθείται από µια τελεία

Παράδειγµα κατηγορήµατος µε γεγονότα

parent(kim, holly).
parent(margaret, kim).
parent(margaret, kent).
parent(esther, margaret).
parent(herbert, margaret).
parent(herbert, jean).

• Ένα πρόγραµµα Prolog µε έξι γεγονότα

• Το οποίο ορίζει ένα κατηγόρηµα (predicate) parent/2

• Αποτελούµενο από γεγονότα που έχουν την «προφανή» 

ερµηνεία: ο kim είναι ο γονιός της holly , η margaret

είναι ο γονιός του kim , κοκ.

Ο διερµηνέας της Prolog

% swipl

Welcome to SWI-Prolog (Multi-threaded, Version 5.2. 13)
Copyright (c) 1990-2003 University of Amsterdam.
SWI-Prolog comes with ABSOLUTELY NO WARRANTY. ...

Please visit http://www.swi-prolog.org for details

For help, use ? - help(Topic). or ? - apropos(Word).

• Το ?- είναι η προτροπή της Prolog για µια ερώτηση

• Ο διερµηνέας είναι διαδραστικός: δέχεται µια ερώτηση, 

προσπαθεί να την απαντήσει, τυπώνει την απάντησή της, 

και επαναλαµβάνει αυτή τη διαδικασία

For help, use ? - help(Topic). or ? - apropos(Word).

?-

Το κατηγόρηµα consult/1

• Κατηγόρηµα του συστήµατος µε το οποίο φορτώνουµε 

?- consult(relations).
% relations compiled 0.00 sec, ΧΧ bytes

Yes
?-

ένα πρόγραµµα από ένα αρχείο στη βάση της Prolog

• Το αρχείο relations (ή relations.pl ) περιλαµβάνει 

τα γεγονότα του κατηγορήµατος parent/2

?- [relations].
% relations compiled 0.00 sec, ΧΧ bytes

Yes



Απλές ερωτήσεις

?- parent(margaret, kent).

Yes
?- parent(fred, pebbles).

No
?-

• Μια ερώτηση προκαλεί την Prolog να αποδείξει κάτι

• Η απάντηση των απλών ερωτήσεων είναι ένα Yes ή No

(Κάποιες ερωτήσεις, όπως η ερώτηση στο κατηγόρηµα

consult/1 , εκτελούνται µόνο για τις παρενέργειές τους)

Η τελεία

• Οι ερωτήσεις µπορούν να εκτείνονται σε πολλές 

?- parent(margaret, kent)
|    .

Yes
?-

• Οι ερωτήσεις µπορούν να εκτείνονται σε πολλές 

γραµµές

• Εάν ξεχάσετε τη τελεία (το χαρακτήρα τέλους της 

ερώτησης), ο διερµηνέας της Prolog θα σας ζητήσει 

περισσότερη είσοδο µε το χαρακτήρα |

Ερωτήσεις µε µεταβλητές

• Οι ερωτήσεις µπορεί να είναι οποιοσδήποτε όρος Prolog, 

συµπεριλαµβανοµένων όρων µε µεταβλητές

• Ο διερµηνέας της Prolog επιστρέφει ως απάντηση τις 

τιµές των µεταβλητών για τις οποίες η ερώτηση µπορεί 

να απαντηθείνα απαντηθεί

?- parent(P, jean).

P = herbert 

Yes
?- parent(P, esther).

No

Στο σηµείο αυτό, ο 
διερµηνέας περιµένει 
κάποια είσοδο.  
∆ίνουµε enter για να 
σταµατήσει η εκτέλεση 
της ερώτησης.

Ευελιξία στις ερωτήσεις

• Οι µεταβλητές µπορούν φυσικά να βρίσκονται σε 

οποιαδήποτε θέση κάποιας ερώτησης:

– parent(Parent, jean)

– parent(esther, Child)

– parent(Parent, Child)parent(Parent, Child)

– parent(Person, Person)

Συζεύξεις (Conjunctions)

?- parent(margaret, X), parent(X, holly).

X = kim 

Yes

• Μια σύζευξη ερωτήσεων είναι µια ακολουθία από όρους 

που διαχωρίζονται από κόµµατα (and)

• Ο διερµηνέας της Prolog προσπαθεί να τις αποδείξει 

όλες (χρησιµοποιώντας το ίδιο σύνολο «δεσιµάτων» για 

τις µεταβλητές της ερώτησης)

Πολλαπλές λύσεις

?- parent(margaret, Child).

Child = kim ;

Child = kent ;

No

• Μπορεί να υπάρχουν περισσότεροι από ένας τρόποι για 

να απαντηθεί µια ερώτηση

• Με το να πληκτρολογήσουµε ; αντί για enter , 

προτρέπουµε το σύστηµα Prolog να βρει και να µας 

δώσει περισσότερες απαντήσεις

No



?- parent(Parent,kim), parent(Grandparent,Parent).

Parent = margaret
Grandparent = esther ;

Parent = margaret
Grandparent = herbert ;

No
?- parent(esther, Child),

?- parent(esther, Child),
|  parent(Child, Grandchild),
|  parent(Grandchild, GreatGrandchild).

Child = margaret
Grandchild = kim
GreatGrandchild = holly 

Yes

Ανάγκη για κανόνες

• Στο προηγούµενο παράδειγµα χρησιµοποιήσαµε µια 

µακροσκελή ερώτηση για τα δισέγγονα της esther

• Θα ήταν καλύτερο εάν µπορούσαµε να γράψουµε κατ’ 

ευθείαν την ερώτηση:

greatgrandparent(esther, GGC)greatgrandparent(esther, GGC)

• Αλλά δε θέλουµε να εισαγάγουµε ξεχωριστά γεγονότα 

αυτού του κατηγορήµατος στη βάση των δεδοµένων 

διότι η σχέση greatgrandparent/2 προκύπτει (και 

πρέπει να προκύπτει) από την ήδη ορισµένη σχέση 

parent/2

Ένα παράδειγµα κανόνα

greatgrandparent(GGP,GGC):-
parent(GGP,GP), 
parent(GP,P),
parent(P,GGC).

συνθήκες

κεφαλή

• Ένας κανόνας καθορίζει πώς αποδεικνύουµε κάτι: για να 

αποδείξουµε την κεφαλή του κανόνα πρέπει να αποδείξουµε 

τις συνθήκες του

• Για να αποδείξουµε greatgrandparent(GGP,GGC) , πρέπει 

να βρούµε τιµές GPκαι P για τις οποίες µπορούµε να 

αποδείξουµε ότι parent(GGP,GP) , µετά ότι parent(GP,P)

και τελικά ότι parent(P,GGC)

Πρόγραµµα µε χρήση του κανόνα

parent(kim,holly).
parent(margaret,kim).
parent(margaret,kent).
parent(esther,margaret).
parent(herbert,margaret).
parent(herbert,jean).

• Ένα πρόγραµµα αποτελείται από µια ακολουθία από 

προτάσεις (clauses)

• Μια πρόταση είναι είτε ένα γεγονός ή ένας κανόνας

greatgrandparent(GGP,GGC) :-
parent(GGP,GP),
parent(GP,P), parent(P,GGC).

Παράδειγµα

• Για την απάντηση της παραπάνω ερώτησης, εσωτερικά 

?- greatgrandparent(esther, GreatGrandchild).

GreatGrandchild = holly ;

No

• Για την απάντηση της παραπάνω ερώτησης, εσωτερικά 

εξετάζονται διάφοροι ενδιάµεσοι στόχοι (goals):

– Ο πρώτος (υπο)στόχος είναι η αρχική ερώτηση

– Ο επόµενος είναι ότι αποµένει προς απόδειξη µετά την απόδειξη 
του πρώτου (υπο)στόχου µε χρήση κάποιας πρότασης του 
προγράµµατος (στη συγκεκριµένη περίπτωση, µε χρήση του 
κανόνα για το κατηγόρηµα  greatgrandparent/2 )

– Κοκ., µέχρι να µη µείνει τίποτα προς απόδειξη

greatgrandparent(esther,GreatGrandchild)

1.   parent(kim,holly).
2.   parent(margaret,kim).
3.   parent(margaret,kent).
4.   parent(esther,margaret).
5.   parent(herbert,margaret).
6.   parent(herbert,jean).
7.   greatgrandparent(GGP,GGC) :-

parent(GGP,GP), parent(GP,P), parent(P,GGC).

Clause 7, binding GGPto esther and GGCto GreatGrandChild

parent(esther,GP), parent(GP,P), parent(P,GreatGrandchild)

parent(margaret,P), parent(P,GreatGrandchild)

parent(kim,GreatGrandchild) 

Clause 7, binding GGPto esther and GGCto GreatGrandChild

Clause 4, binding GPto margaret

Clause 2, binding P to kim

Clause 1, binding GreatGrandchild to holly



Κανόνες µε χρήση άλλων κανόνων

• Η ίδια σχέση µε την προηγούµενη, ορισµένη σε στρώσεις

grandparent(GP,GC) :-
parent(GP,P), parent(P,GC).

greatgrandparent(GGP,GGC) :-
grandparent(GGP,P), parent(P,GGC).

• Η ίδια σχέση µε την προηγούµενη, ορισµένη σε στρώσεις

• Παρατηρήστε ότι και οι δύο κανόνες χρησιµοποιούν 

µεταβλητές µε όνοµα P

• Η εµβέλεια κάποιας µεταβλητής είναι µόνο ο κανόνας 

που την περιέχει

• Οι δύο µεταβλητές µε όνοµα P είναι ανεξάρτητες

Αναδροµικοί κανόνες

• Ένας X είναι πρόγονος κάποιου Y:

ancestor(X,Y) :- parent(X,Y).
ancestor(X,Y) :-

parent(Z,Y),
ancestor(X,Z).

• Ένας X είναι πρόγονος κάποιου Y:

– Εάν ο X είναι γονιός του Y, ή 

– Εάν υπάρχει κάποιος Z τέτοιος ώστε ο Z να είναι ο γονιός του Y, 
και ο X να είναι πρόγονος του Z

• Η Prolog δοκιµάζει τους κανόνες µε τη σειρά που αυτοί 

εµφανίζονται, οπότε είναι καλό (και µάλιστα πολλές 

φορές αναγκαίο) τα γεγονότα και οι µη αναδροµικοί 

κανόνες να εµφανίζονται πριν από τους αναδροµικούς

?- ancestor(jean, jean).

No
?- ancestor(kim, holly).

Yes
?- ancestor(A, holly).

A = kim ;

A = margaret ;

A = esther ;

A = herbert ;

No

Ο πυρήνας της σύνταξης της Prolog

• Ολόκληρη η σύνταξη του πυρήνα της Prolog:

<clause> ::=  <fact> | <rule>
<fact> ::=  <term> .
<rule> ::=  <term> :- <termlist> .
<termlist> ::=  <term> | <term> , <termlist>

• Εάν εξαιρεθεί η δυνατότητα ορισµού νέων τελεστών από 

το χρήστη, συντακτικά η Prolog είναι πολύ απλή γλώσσα

• Αυτό ήταν όλο!

Η διαδικαστική όψη της Prolog

• Ένας κανόνας λέει πως µπορούµε να αποδείξουµε κάτι:

greatgrandparent(GGP,GGC) :-
parent(GGP,GP),  
parent(GP,P),
parent(P,GGC).

• Ένας κανόνας λέει πως µπορούµε να αποδείξουµε κάτι:

– Για να αποδείξουµε greatgrandparent(GGP,GGC) , πρέπει να 
βρούµε τιµές GPκαι P για τις οποίες µπορούµε να αποδείξουµε
ότι parent(GGP,GP) , µετά ότι parent(GP,P) και τελικά ότι
parent(P,GGC)

• Ένα πρόγραµµα Prolog προσδιορίζει διαδικασίες 

εύρεσης απαντήσεων σε ερωτήσεις

Εισαγωγή στη γλώσσα Prolog 29

Η δηλωτική όψη της Prolog

• Ένας κανόνας είναι µια λογική δήλωση:

– Για όλες τις τιµές των GGP, GP, P, και GGC, αν ισχύει ότι 
parent(GGP,GP) και parent(GP,P) και parent(P,GGC) , 
τότε επίσης ισχύει και greatgrandparent(GGP,GGC)

• Με άλλα λόγια, ένας κανόνας είναι απλώς µια φόρµουλα

– δε λέει πώς θα γίνει κάτι αλλά απλώς ορίζει τις προϋποθέσεις – δε λέει πώς θα γίνει κάτι αλλά απλώς ορίζει τις προϋποθέσεις 
κάτω από τις οποίες µια πρόταση είναι αληθής:

∆ηλωτικές γλώσσες προγραµµατισµού

• Κάθε κοµµάτι του προγράµµατος αντιστοιχεί σε µια 

απλή µαθηµατική έννοια:

– Προτάσεις της Prolog – σε φόρµουλες της κατηγορηµατικής 
λογικής πρώτης τάξης (first-order predicate logic)

– Οι ορισµοί συναρτήσεων στην ML – σε µαθηµατικές συναρτήσεις

• Πολλές φορές ο όρος δηλωτικές (declarative) γλώσσες• Πολλές φορές ο όρος δηλωτικές (declarative) γλώσσες

προγραµµατισµού  χρησιµοποιείται για να δείξει την 

αντιδιαστολή των γλωσσών αυτών σε σχέση µε τις 

προστακτικές (imperative) γλώσσες

• Ως δηλωτικές γλώσσες συνήθως θεωρούµε τις λογικές 

και τις συναρτησιακές

Πλεονεκτήµατα των δηλωτικών γλωσσών

• Τα προγράµµατα σε προστακτικές γλώσσες πολλές 

φορές έχουν προβλήµατα λόγω παρενεργειών και 

αλληλεξαρτήσεων

• Επειδή οι δηλωτικές γλώσσες έχουν σχετικά απλή και 

«καθαρότερη» σηµασιολογία, αποφεύγονται κάποιου «καθαρότερη» σηµασιολογία, αποφεύγονται κάποιου 

είδους προγραµµατιστικά λάθη µε συνέπεια η ανάπτυξη 

και η συντήρηση προγραµµάτων να γίνεται πιο εύκολα

• Ο προγραµµατισµός είναι ανώτερου επιπέδου και πιο 

αυτόµατος: ο προγραµµατιστής απλώς περιγράφει την 

προδιαγραφή του προβλήµατος και κάποια κοµµάτια της 

εκτέλεσής του παρέχονται από το σύστηµα υλοποίησης

Εισαγωγή στη γλώσσα Prolog 32

Η Prolog έχει και τα δύο χαρακτηριστικά

• Έχει δηλωτική σηµασιολογία

– Ένα πρόγραµµα Prolog έχει ένα λογικό περιεχόµενο

• Έχει διαδικαστική σηµασιολογία

– Ένα πρόγραµµα Prolog έχει και καθαρά διαδικαστικά 
χαρακτηριστικά: τη σειρά µε την οποία εµφανίζονται οι 
κανόνες, τη σειρά εµφάνισης των υποστόχων σ’ ένα κανόνες, τη σειρά εµφάνισης των υποστόχων σ’ ένα 
κανόνα, κατηγορήµατα µε παρενέργειες (π.χ. Ι/Ο), κ.λπ.

• Όµως είναι σηµαντικό ο προγραµµατιστής να είναι 

ενήµερος και να παίρνει υπόψη του και τις δύο όψεις της 

γλώσσας κατά τον προγραµµατισµό

Τελεστές της Prolog

• Η Prolog έχει κάποιους προκαθορισµένους τελεστές

(και επιτρέπει τον ορισµό νέων τελεστών από το χρήστη)

• Ένας τελεστής είναι απλώς ένα κατηγόρηµα για το 

οποίο είναι δυνατή η χρήση κάποιας ειδικής συντακτικής 

συντοµογραφίας στις χρήσεις τουσυντοµογραφίας στις χρήσεις του

Το κατηγόρηµα =/2

• Ο στόχος =(X,Y) επιτυγχάνει αν και µόνον αν οι όροι X

και Y µπορούν να ενοποιηθούν:

?- =(parent(adam,seth), parent(adam,X)).

X = seth ;

• Επειδή το άτοµο '=' είναι ορισµένο ως δυαδικός 

τελεστής, συνήθως γράφουµε το παραπάνω ως:

No

?- parent(adam,seth) = parent(adam,X).

X = seth ;

No

Αριθµητικοί τελεστές

• Τα άτοµα '+' , '-' , '*' και '/' είναι ορισµένα ως 

τελεστές, µε τη συνήθη προτεραιότητα και 

προσεταιριστικότητα

?- X = +(1,*(2,3)).
Η Prolog µας επιτρέπει να 

X = 1+2*3 ;

No
?- X = 1+2*3.

X = 1+2*3 ;

No

Η Prolog µας επιτρέπει να 
χρησιµοποιήσουµε σύνταξη 
τελεστών στην είσοδο, και 
χρησιµοποιεί αυτή τη µορφή 
στην έξοδο των όρων, αλλά ο 
υποκείµενος όρος είναι ο
+(1,*(2,3))



Μη αποτιµηµένοι όροι

?- +(X,Y) = 1+2*3.

X = 1
Y = 2*3 ;

No
?- 7 = 1+2*3.

• Μετά την ενοποίηση, ο όρος παραµένει +(1,*(2,3))

• Με άλλα λόγια, ο όρος είναι µη αποτιµηµένος

(Σε επόµενη διάλεξη θα δούµε ότι υπάρχει τρόπος στην 

Prolog να αποτιµήσουµε τέτοιους όρους)

No

Λίστες στην Prolog

• Συντακτικά είναι περίπου σαν τις λίστες στην ML

• Το άτοµο [] αναπαριστά την κενή λίστα

• Το συναρτησιακό σύµβολο '.'/2 αντιστοιχεί στον 

τελεστή :: της ML

Έκφραση στην ML Όρος στην Prolog
[] []

1::[] .(1,[])

1::2::3::[] .(1,.(2,.(3,[])))

∆εν υπάρχει αντίστοιχο. .(1,.(parent(X,Y),[]))

Οι αναπαραστάσεις των λιστών στην Prolog

Λίστα Όρος στον οποίο αντιστοιχεί
[] []

[1] .(1,[])

[1,2,3] .(1,.(2,.(3,[])))

[1,parent(X,Y)] .(1,.(parent(X,Y),[]))

• Η µορφή που συνήθως χρησιµοποιείται είναι η αριστερή

• Αλλά είναι απλώς µια συντοµογραφία της µορφής που 

φαίνεται δεξιά στον παραπάνω πίνακα

• Η Prolog τυπώνει τις λίστες µε χρήση της µορφής στο 

αριστερό µέρος του πίνακα

Παράδειγµα

?- X = .(1,.(2,.(3,[]))).

X = [1, 2, 3] ;

No
?- .(X,Y) = [1,2,3].

X = 1

X = 1
Y = [2, 3] ;

No

Αναπαραστάσεις λιστών µε ουρά

• Πολλές φορές το τέλος της λίστας είναι το σύµβολο |

ακολουθούµενο από την ουρά της λίστας

Λίστα Όρος στον οποίο αντιστοιχεί
[1|X] .(1,X)

[1,2|X] .(1,.(2,X)))

[1,2|[3,4]] Το ίδιο µε [1,2,3,4]

ακολουθούµενο από την ουρά της λίστας

• Χρησιµοποιείται σε πρότυπα: ο όρος [1,2|X] ενοποιείται 

µε κάθε λίστα που αρχίζει µε 1,2 και έχει έναν όρο X ως 

ουρά

?- [1,2|X] = [1,2,3,4,5].

X = [3, 4, 5] ;
No

Το κατηγόρηµα append/3

• Το προκαθορισµένο κατηγόρηµα append(X,Y,Z)

επιτυγχάνει αν και µόνο αν η λίστα Z είναι το αποτέλεσµα 

της συνένωσης της λίστας Y στο τέλος της λίστας X

?- append([1,2],[3,4],Z).

Z = [1, 2, 3, 4] ;

Z = [1, 2, 3, 4] ;

No



Όχι απλώς µια συνάρτηση!

?- append(X, [3,4], [1,2,3,4]).

X = [1, 2] ;

No

• Το κατηγόρηµα append µπορεί να χρησιµοποιηθεί µε 

κάθε όρο της Prolog (µε άλλα λόγια, µε µεταβλητές σε 

κάθε όρισµα)

Όχι απλώς µια συνάρτηση!

?- append(X, Y, [1,2,3]).

X = []
Y = [1, 2, 3] ;

X = [1]
Y = [2, 3] ;

X = [1, 2]
Y = [3] ;

X = [1, 2, 3]
Y = [] ;

No

Η υλοποίηση του κατηγορήµατος append/3

append([], L, L).
append([H|L1], L2, [H|L3]) :-

append(L1, L2, L3).

Άλλα κατηγορήµατα για λίστες

Κατηγόρηµα Περιγραφή

member(X,Y) Επιτυγχάνει εάν η λίστα Y έχει ως µέλος τον όρο X.

select(X,Y,Z)
Επιτυγχάνει εάν η λίστα Y περιέχει το X ως µέλος, και 
ο όρος Z είναι ο ίδιος µε τη λίστα Y αλλά χωρίς ένα 
στιγµιότυπο του όρου X.

nth0(N,L,X) Επιτυχάνει εάν ο N είναι ένας µη αρνητικός ακέραιος, 

• Όλα τα παραπάνω είναι ευέλικτα, σαν το append/3

• Με άλλα λόγια οι ερωτήσεις στα κατηγορήµατα µπορούν 

να περιέχουν µεταβλητές σε οποιαδήποτε όρισµά τους

nth0(N,L,X) Επιτυχάνει εάν ο N είναι ένας µη αρνητικός ακέραιος, 
L µια λίστα, και το X είναι το N-οστό στοιχείο της 
λίστας L, αρχίζοντας την αρίθµηση από το 0.

length(L,N) Επιτυγχάνει εάν ο όρος L είναι µια λίστα µε µήκος N.

Χρήση του κατηγορήµατος select/3

?- select(2, [1,2,3], Z).

Z = [1, 3] ;

No

?- select(2, [1,2,3,2], Z).

Z = [1, 3, 2] ;

Z = [1, 2, 3] ;

No
?- select(2, Y, [1,3]).

?- select(2, Y, [1,3]).

Y = [2, 1, 3] ;

Y = [1, 2, 3] ;

Y = [1, 3, 2] ;

No

?- select(2, [ Χ,Y], [1]).

X = 2
Y = 1 ;

X = 1
Y = 2 ;

No

Το κατηγόρηµα reverse/2

• reverse(X,Y) είναι αληθές εάν ο όρος Y ενοποιείται 

µε την αναστροφή της λίστας X

?- reverse([1,2,3,4], Y).

Y = [4, 3, 2, 1] ;

No



Η υλοποίηση του κατηγορήµατος reverse/2

• ∆εν είναι ένας αποδοτικός τρόπος να αντιστρέψουµε µια 

reverse([], []).
reverse([Head|Tail], Reversed) :-

reverse(Tail, RevTail),
append(RevTail, [Head], Reversed).

• ∆εν είναι ένας αποδοτικός τρόπος να αντιστρέψουµε µια 

λίστα

• Υπάρχουν πιο αποδοτικοί τρόποι αντιστροφής λιστών

(όπως ακριβώς είδαµε και στην ML)

Όταν δεν πάνε όλα καλά…

?- reverse(L, [1,2,3,4]).

L = [4, 3, 2, 1] ;

Action (h for help) ? a
% Execution Aborted
?-

• Ζητήσαµε µια ακόµα λύση και βρεθήκαµε σ’ έναν 

ατέρµονο βρόχο

• Μπορούµε να παύσουµε την εκτέλεση πατώντας Ctrl-C , 

και στη συνέχεια πατάµε a για το οριστικό σταµάτηµά της

• Το reverse/2 δεν είναι τόσο ευέλικτο όσο το append/3

Αναστρέψιµα και µη αναστρέψιµα κατηγορήµατα

• Σ’ έναν ιδεατό κόσµο, τα κατηγορήµατα είναι όλα τόσο 

ευέλικτα όσο το append/3

• Είναι πιο δηλωτικά και µε λιγότερους περιορισµούς στη 

χρήση τους

• Αλλά πολλές φορές µη ευέλικτες υλοποιήσεις είναι • Αλλά πολλές φορές µη ευέλικτες υλοποιήσεις είναι 

προτιµότερες, για λόγους απόδοσης ή/και απλότητας

• Ένα παράδειγµα του τι σηµαίνει η παραπάνω φράση µας 

δείχνει το ενσωµατωµένο κατηγόρηµα sort/2

Το κατηγόρηµα sort/2

?- sort([2,3,1,4], X).

X = [1, 2, 3, 4] ;

No
?- sort(X, [1,2,3,4]).
ERROR: Arguments are not sufficiently instantiated

• Ένα αναστρέψιµο κατηγόρηµα sort/2 θα έπρεπε να 

µπορεί να «αντι-ταξινοµήσει» µια ήδη ταξινοµηµένη 

λίστα — µε άλλα λόγια θα έπρεπε να επιστρέφει όλες τις 

αναδιατάξεις των στοιχείων της — που είναι εκθετικό

• Θέλουµε κάτι σαν το παραπάνω να είναι ενσωµατωµένο 

στη γλώσσα;

Ανώνυµες µεταβλητές

• Η µεταβλητή _ είναι µια µεταβλητή χωρίς όνοµα

• Κάθε εµφάνισή της είναι ανεξάρτητη από όλες τις 

υπόλοιπες

• Ενοποιείται µε κάθε όρο

• Με άλλα λόγια, συµπεριφέρεται σαν τις µεταβλητές• Με άλλα λόγια, συµπεριφέρεται σαν τις µεταβλητές _

της ML

Εισαγωγή στη γλώσσα Prolog 53

Το κατηγόρηµα member/2

member(X, [X|_]).
member(X, [_|T]) :-

member(X, T).

?- member(b, [a,b,c]).

Yes

Yes

?- member(d, [a,b,c]).

No

?- member(X, [a,b]).

X = a ;

X = b ;

No



Προσοχή στις προειδοποιήσεις!

append([], L, L).
append([H|L1], L2, [H|L3]) :-

append(LI, L2, L3).

Warning: Singleton variables [L1,LI]

• Μην αγνοείτε τα προειδοποιητικά µηνύµατα ότι 

µεταβλητές εµφανίζονται µόνο µια φορά

• Όπως και στην ML, είναι κακό στυλ προγραµµατισµού να 

ονοµάζουµε µεταβλητές που δεν χρησιµοποιούµε – ειδικά 

επειδή είναι το µόνο µήνυµα που ο διερµηνέας θα βγάλει

Το κατηγόρηµα \+/1

?- member(1, [1,2,3]).

Yes
?- \+ member(4, [1,2,3]).

Yes

• Σε απλές εφαρµογές, συνήθως λειτουργεί σαν τον 

τελεστή λογικής άρνησης

• Αλλά έχει και µια σηµαντική διαδικαστική πλευρά…

Yes

Άρνηση ως αποτυχία (negation as failure)

• Για να αποδείξει το \+ X , η Prolog προσπαθεί να 

αποδείξει το X

• \+ X επιτυγχάνει εάν ο στόχος X αποτυγχάνει

• Οι δύο όψεις ξανά:

– ∆ηλωτική∆ηλωτική∆ηλωτική∆ηλωτική: \ + X = ¬X– ∆ηλωτική∆ηλωτική∆ηλωτική∆ηλωτική: \ + X = ¬X

– ∆ιαδικαστική∆ιαδικαστική∆ιαδικαστική∆ιαδικαστική: \+ X επιτυγχάνει εάν το X αποτυγχάνει, και 
αποτυγχάνει εάν το X επιτυγχάνει, και δεν τερµατίζει εάν 
το X δεν τερµατίζει

sibling(X, Y) :-
parent(P, X), 
parent(P, Y),
\+ (X = Y).

?- sibling(X, Y).

X = kim
Y = kent ;

X = kent

sibling(X, Y) :-
\+ (X = Y), 
parent(P, X), 
parent(P, Y).

?- sibling(kim, kent).

Παράδειγµα

X = kent
Y = kim ;

X = margaret
Y = jean ;

X = jean
Y = margaret ;

No

Yes
?- sibling(kim, kim).

No
?- sibling(X, Y).

No

Ένα µεγαλύτερο παράδειγµα σε Prolog

Μια κλασική σπαζοκεφαλιά

• Ένας άνθρωπος (man) ταξιδεύει µαζί µ’ ένα λύκο (wolf), 

µια κατσίκα (goat) και ένα λάχανο (cabbage)

• Ο άνθρωπος θέλει να διασχίσει ένα ποτάµι από τη δυτική 

όχθη στην ανατολική

• Στην αρχική όχθη υπάρχει µια βάρκα, η οποία όµως • Στην αρχική όχθη υπάρχει µια βάρκα, η οποία όµως 

χωράει τον άνθρωπο και το πολύ ένα άλλο αντικείµενο

• Ο λύκος τρώει την κατσίκα εάν βρεθεί µόνος µαζί της

• Η κατσίκα τρώει το λάχανο εάν βρεθεί µόνη µαζί του

• Με ποιο τρόπο µπορεί ο άνθρωπος να περάσει στην 

απέναντι όχθη µε όλα τα υπάρχοντά του;

∆οµές δεδοµένων της λύσης

• Θα αναπαραστήσουµε τους σχηµατισµούς του 

συστήµατος µε µια λίστα που δείχνει την όχθη στην 

οποία είναι το κάθε τι µε την εξής σειρά:

man, wolf, goat, cabbage

• Ο αρχικός σχηµατισµός: [w,w,w,w]• Ο αρχικός σχηµατισµός: [w,w,w,w]

• Εάν για παράδειγµα ο άνθρωπος µε το λύκο µπουν στη 

βάρκα, ο νέος σχηµατισµός είναι ο [e,e,w,w] – αλλά 

τότε η κατσίκα θα φάει το λάχανο,οπότε ο παραπάνω 

σχηµατισµός δεν είναι επιτρεπτός ως ενδιάµεσος

• Ο επιθυµητός τελικός σχηµατισµός: [e,e,e,e]

Κινήσεις

• Σε κάθε κίνηση, ο άνθρωπος διασχίζει το ποτάµι µε το 

πολύ ένα από τα υπάρχοντά του

• Θα αναπαραστήσουµε τις τέσσερις επιτρεπτές κινήσεις 

µε τέσσερα άτοµα: wolf , goat , cabbage , nothing

• (Όπου, nothing σηµαίνει ότι ο άνθρωπος διασχίζει το • (Όπου, nothing σηµαίνει ότι ο άνθρωπος διασχίζει το 

ποτάµι µόνος του µέσα στη βάρκα)

Οι κινήσεις τροποποιούν τον σχηµατισµό 

• Κάθε κίνηση µεταµορφώνει ένα σχηµατισµό σ’ έναν άλλο

• Στην Prolog, θα προγραµµατίσουµε τη λύση µε ένα 

κατηγόρηµα: move(Config, Move, NextConfig)

– Config είναι ένας σχηµατισµός (π.χ. [w,w,w,w] )

– Move είναι µια κίνηση (π.χ. wolf )– Move είναι µια κίνηση (π.χ. wolf )

– NextConfig είναι ο σχηµατισµός-αποτέλεσµα (στο συγκεκριµένο 
παράδειγµα ο [e,e,w,w] )

Το κατηγόρηµα move/2

change(e, w) .
change(w, e).

move([X,X,Goat,Cabbage], wolf, [Y,Y,Goat,Cabbage]) :-
change(X, Y).

move([X,Wolf,X,Cabbage], goat, [Y,Wolf,Y,Cabbage]) :-
change(X, Y).

change(X, Y).
move([X,Wolf,Goat,X], cabbage, [Y,Wolf,Goat,Y]) :-

change(X, Y).
move([X,Wolf,Goat,C], nothing, [Y,Wolf,Goat,C]) :-

change(X, Y).

Ασφαλείς σχηµατισµοί

• Ένας σχηµατισµός είναι ασφαλής εάν

– Τουλάχιστον ένας από τους λύκο και κατσίκα είναι στην ίδια 
όχθη µε τον άνθρωπο, και

– Τουλάχιστον ένας από τους κατσίκα και λάχανο είναι στην ίδια 
όχθη µε τον άνθρωπο

oneEq(X, X, _).
oneEq(X, _, X).

safe([Man, Wolf, Goat, Cabbage]) :-
oneEq(Man, Goat, Wolf),
oneEq(Man, Goat, Cabbage).

Λύσεις

• Μια λύση αρχίζει από τον αρχικό σχηµατισµό και 

δηµιουργεί µια λίστα από κινήσεις που οδηγούν στο 

σχηµατισµό [e,e,e,e] , έτσι ώστε όλοι οι ενδιάµεσοι 

σχηµατισµοί να είναι ασφαλείς

solution([e,e,e,e], []).

solution([e,e,e,e], []).
solution(Config, [Move|Moves]) :-

move(Config, Move, NextConfig),
safe(NextConfig),
solution(NextConfig, Moves).



Το κατηγόρηµα length/2

?- length(L, N).

X = []
Y = 0 ;

X = [_]
Y = 1 ;

X = [_, _]
Y = 2 ;

X = [_, _, _]
Y = 3 ;

X = [_, _, _, _]
Y = 4 

Yes

Η Prolog βρίσκει τη λύση

ΣηµείωσηΣηµείωσηΣηµείωσηΣηµείωση: χωρίς τη χρήση του length /2 , η Prolog δε θα 

?- length(L, Ν), solution([w,w,w,w], L).

L = [goat, nothing, wolf, goat, cabbage, nothing, g oat]
Ν = 7

Yes

ΣηµείωσηΣηµείωσηΣηµείωσηΣηµείωση: χωρίς τη χρήση του length /2 , η Prolog δε θα 

έβρισκε κάποια λύση (µε το συγκεκριµένο πρόγραµµα). 

Θα µπερδευόταν ψάχνοντας για λύσεις της µορφής: 

[goat,goat,goat,goat,goat…]

Κάποια καλά χαρακτηριστικά της Prolog

• Το πρόγραµµά µας απλώς κατέγραψε την εκφώνηση του 

προβλήµατος προς επίλυση

• ∆εν προσδιορίσαµε κάποιο τρόπο αναζήτησης της λύσης 

του προβλήµατος  – η ίδια η Prolog έκανε την αναζήτηση

• Αυτού του είδους τα προβλήµατα είναι ακριβώς τα • Αυτού του είδους τα προβλήµατα είναι ακριβώς τα 

προβλήµατα που είναι κατάλληλα προς επίλυση σε

Prolog

• Περισσότερα παραδείγµατα σε επόµενες διαλέξεις…

Εισαγωγή στη γλώσσα Prolog 69


